
The Airspace System

The Airspace System

I 24h of air traffic in the USA

I 24h of air traffic in the World

The Airspace System

Next Generation Air Transportation System (NextGen)

I FAA and industry forecast a 1.5 – 2.5 times increase of air
traffic operations over the next two decades.

I By 2016, develop and demonstrate future concepts,
capabilities, and technologies that will enable major increases
in air traffic management . . . while maintaining safety, to meet
capacity and mobility requirement of NextGen.

The Airspace System

Enabling Technology

I Hardware Systems:
I Global Positioning System (GPS).
I Automatic Dependent Surveillance-Broadcast (ADS-B).
I Increase in computational power.

I Operational concepts based on distributed and autonomous
systems.

The Airspace System

Distributed Air/Ground Air Traffic Management

I Today: Air traffic controllers have the primary responsibility
for en-route aircraft separation

I Future: The responsibility for separation will be air/ground
distributed.

Automated Cockpit

The Airspace System

State-Based Separation Assurance Systems (SAS)

Airborne systems that maintain aircraft separated assuming only
state information (current position and velocity vector), e.g.,
Conflict Detection and Resolution (CD&R).

The Airspace System

SAS Safety Properties

I Conflict Detection:
I correctness: No false alerts.
I completeness: No missed alerts.

I Conflict Resolution:
I completeness: At least one solution.
I independence: Conflict solved when one aircraft maneuvers.
I (Implicit) coordination: Conflict solved when both aircraft

maneuver.

The Airspace System

Airborne Coordinated Conflict Resolution and Detection
Framework

ACCoRD is a mathematical framework, fully developed in PVS, for
the design and verification of state-based separation assurance
algorithms.

2D Conflict Detection and Resolution (CDR2D)

I 2D Rectangular coordinate system.

I Linear trajectories.

I Instantaneous velocity changes.

I Perfect state information (position and velocity vector).

I Perfect communication (broadcast from traffic aircraft).

I Pairwise approach: ownship and intruder aircraft.

I Distributed approach: No central control.

I Independent decision making: aircraft do not
communicate/negotiate their resolution maneuvers.

2D Conflict Detection and Resolution (CDR2D)

Basic Definitions: Protected Zone
Circular area of radius D around each aircraft. The value D defines
a minimum horizontal distance between aircraft.

D

(Typically, D = 5nm)

2D Conflict Detection and Resolution (CDR2D)

Basic Definitions: Loss of Separation

Violation of separation minima, i.e., overlapping of protected zones.

2D Conflict Detection and Resolution (CDR2D)

Basic Definitions: Conflict
Predicted loss of separation within a lookahead time T .

T

(Typically, T = 5min)

2D Conflict Detection and Resolution (CDR2D)

Geometry

I Ownship current state is given by so (postion) and vo

(velocity) in R2.

I Intruder current state is given by si (postion) and vi (velocity)
in R2.

I Projected states at time t: so + t vo and si + t vi .

I Relative view: s ≡ so − si and v ≡ vo − vi .

2D Conflict Detection and Resolution (CDR2D)

Transformation to Relative Coordinate System
In the relative coordinate system, the intruder fixed at the origin
and the ownship moves relative to the intruder.

2D Conflict Detection and Resolution (CDR2D)

Conflict

Given a relative position s and relative velocity v, there is a
(predicted) conflict if there exists a time 0 ≤ t ≤ T such that the
aircraft are in loss of separation at time t:

conflict?(s, v) ≡ ∃ t ∈ [0,T ] : ‖s + t v‖ < D.

2D Conflict Detection and Resolution (CDR2D)

Conflict Detection Algorithms

A conflict detection algorithm is a function

cd(so , vo , si , vi):bool,

where

I so , vo is current state of the ownship.

I si , vi is the current state of the intruder.

2D Conflict Detection and Resolution (CDR2D)

Conflict Detection Correctness and Completeness

I An algorithm cd is correct if it does not have missed alerts,
i.e.,

conflict?(so − si , vo − vi ) =⇒ cd(so , vo , si , vi ).

I An algorithm cd is complete if it does not have false alerts,
i.e.,

cd(so , vo , si , vi ) =⇒ conflict?(so − si , vo − vi ).

2D Conflict Detection and Resolution (CDR2D)

Conflict Resolution Algorithms

A conflict resolution algorithm is a function

cr(so , vo , si , vi) : ℘(R2),

where v′o ∈ cr(so , vo , si , vi ) is a resolution maneuver for the
ownship.

2D Conflict Detection and Resolution (CDR2D)

Independent Conflict Resolution

An algorithm cr is independent if it provides conflict-free resolution
maneuvers assuming that only the ownship maneuvers, i.e.,

For all v′o ∈ cr(so , vo , si , vi ),

conflict?(so − si , vo − vi ) =⇒ ¬conflict?(so − si , v
′
o − vi ).

2D Conflict Detection and Resolution (CDR2D)

Coordinated Conflict Resolution

Algorithms cro and cri are (implicitly) coordinated if they provide
conflict-free resolution maneuvers assuming that both aircraft
simultaneously maneuver, i.e.,

For all v′o ∈ cro(so , vo , si , vi ), v′i ∈ cri (si , vi , so , vo),

conflict?(so − si , vo − vi ) =⇒ ¬conflict?(so − si , v
′
o − v′i ).

Real Numbers in PVS

Real Numbers in PVS

I Reals are defined as an uninterpreted subtype of number in
the prelude library:

real: TYPE+ FROM number

I All numeric constants are real:
I naturals: 0,1,...
I integers: ...,-1,0,1,...
I rationals: ...,-1/10,...,3/2,...

I Decimal notation is supported: The decimal number
3.141516 is syntactic sugar for the rational number
31416/10000.

Real Numbers in PVS

PVS’s real numbers are Real
(Rather than floating point numbers)

I All the standard properties: infinite, non-enumerable,
N ⊆ Z ⊆ Q ⊆ R, . . . .

I Exact arithmetic: 1/3 + 1/3 + 1/3 = 1.

I The type real is unbounded:

googol : real = 10^100
googolplex : real = 10^googol

googol_prop : LEMMA
googolplex > googol * googol

Real Numbers in PVS

Ground real Arithmetic is Built-in

I Numerical expressions can be automatically reduced by the
theorem prover (no need to prove 1+1=2), . . .

I . . . except for machine physical limitations, e.g., you probably
cannot prove googol prop with (grind).

I But, you can still prove googol prop using analytical
methods.

Real Numbers in PVS

Subtypes of real

nzreal : TYPE+ = {r:real| r /= 0} % Nonzero reals
nnreal : TYPE+ = {r:real| r >= 0} % Nonnegative reals
npreal : TYPE+ = {r:real| r <= 0} % Nonpositive reals
negreal : TYPE+ = {r:real| r < 0} % Negative reals
posreal : TYPE+ = {r:real| r > 0} % Positive reals

rat : TYPE+ FROM real
int : TYPE+ FROM rat
nat : TYPE+ FROM int

The uninterpreted type number is the only real’s supertype
predefined in PVS: no complex numbers, no hyper-reals,
no R∞, . . .

Real Numbers in PVS

Predefined Operations

+, -, *: [real, real -> real]
/: [real, nzreal -> real]
-: [real -> real]

sgn(x:real) : int = IF x >= 0 THEN 1 ELSE -1 ENDIF
abs(x:real) : {nny: nnreal | nny >= x} = ...
max(x,y:real): {z: real | z >= x AND z >= y} = ...
min(x,y:real): {z: real | z <= x AND z <= y} = ...
^(x: real,i:{i:int | x /= 0 OR i >= 0}): real = ...

. . . and what about √,
∫
, log, exp, sin, cos, tan, π, lim, . . . ?

Real Numbers in PVS

NASA PVS Libraries
http://shemesh.larc.nasa.gov/fm/ftp/larc/PVS-library/pvslib.html

I reals: Square, square root, quadratic formula, polynomials.

I analysis: Real analysis, limits, continuity, derivatives,
integrals.

I vectors and vect analysis: Vector calculus and analysis.

I series: Power series, Taylor’s theorem.

I lnexp and lnexp fnd: Logarithm, exponential, and
hyperbolic functions.

I trig and trig fnd: Trigonometry.

I complex: Complex numbers.

I float: Floating point numbers.

I . . .

http://shemesh.larc.nasa.gov/fm/ftp/larc/PVS-library/pvslib.html

Real Numbers in PVS

To Be Or Not To Be (Fundational) ?

I Axiomatic theories trig and lnexp typechek faster.

I Fundational theories trig fnd and lnexp fnd have no
axioms.

I Be careful what you wish for:

|-------
{1} sin(pi / 2) > 1 / 2

Rule? (grind)
Integral rewrites Integral[real](0, 1, atan_deriv_fn)
to integral(0, 1, atan_deriv_fn)

atan_value rewrites atan_value(1)
to integral(0, 1, atan_deriv_fn)

atan rewrites atan(1)
...

Formal Development of CDR2D

CDR2D[D:posreal] : THEORY
BEGIN IMPORTING vectors@vectors_2D
% 2D Positions
Position : TYPE = Vect2
p,s,so,si : VAR Position

% 2D Relative Velocities
Velocity : TYPE = Vect2
v : VAR Velocity

% 2D Absolute Velocities (cannot be zero)
AbsVelocity : TYPE = Nz_vect2
vo,vop,vi,vip : VAR AbsVelocity
...

END CDR2D

Formal Development of CDR2D

Relative Protected Zone

ProtectedZone : set[Position] = {p | sq(p) < sq(D)}

Sets in PVS are charaterictic funtions, e.g., the definition above is
equivalent to

ProtectedZone(p) : bool = sq(p) < sq(D)



Real Applications

Formal Development of CDR2D

Relative Protected Zone

ProtectedZone : set[Position] = {p | sq(p) < sq(D)}

Sets in PVS are charaterictic funtions, e.g., the definition above is
equivalent to

ProtectedZone(p) : bool = sq(p) < sq(D)



Real Applications

Formal Development of CDR2D

Conflict, Conflict Detection, and Conflict Resolution

conflict?(s,v): bool =
EXISTS(t:nnreal|t <= T): ProtectedZone(s+t*v)

% Type of CD algorithms
CD : TYPE =
PRED[[Position,AbsVelocity,Position,AbsVelocity]]

% Type of CR algorithms
CR : TYPE = [[Position,AbsVelocity,Position,AbsVelocity]

-> set[AbsVelocity]]

Note: PRED[T] is the same as [T->bool].



Real Applications

Formal Development of CDR2D

CD Correctness and Completeness

cd : VAR CD

correct?(cd) : bool =
FORALL(so,vo,si,vi):
conflict?(so-si,vo-vi) IMPLIES cd(so,vo,si,vi)

complete?(cd) : bool =
FORALL(so,vo,si,vi):
cd(so,vo,si,vi) IMPLIES conflict?(so-si,vo-vi)

Formal Development of CDR2D

CR Independence and Coordination

cr : VAR CR

independent?(cr) : bool =
FORALL(so,vo,si,vi,vop) :
conflict?(so-si,vo-vi) AND
member(vop,cr(so,vo,si,vi)) IMPLIES
NOT conflict?(so-si,vop-vi)

coordinated?(cr) : bool =
FORALL(so,vo,si,vi,vop,vip) :
conflict?(so-si,vo-vi) AND
member(vop,cr(so,vo,si,vi)) IMPLIES
member(vip,cr(si,vi,si,vi)) IMPLIES
NOT conflict?(so-si,vop-vip)

Formal Development of CDR2D

A CD Algorithm

cd2d_ever?(s,v): MACRO bool =
IF s*v < 0 THEN
sq(s*v)-sqv(v)*(sqv(s)-sq(D)) > 0

ELSE sqv(s) < sq(D)
ENDIF

cd2d_ever(so,vo,si,vi): bool =
cd2d_ever?(so-si,vo-vi)

cd2d_ever_incomplete : THEOREM
NOT complete?(cd2d_ever)

cd2d_ever_correct : THEOREM
correct?(cd2d_ever)

Formal Development of CDR2D

Exercise 1

1. Define a type of aircraft with an identifier, a 2D position, and
a 2D velocity:

AircraftXY : TYPE = ...

2. Define a constant ac of type AircraftXY:

ac : AircraftXY = ...

3. Define a predicate cdnas that holds if there is a traffic aircraft
that is in conflict ever, e.g., cd2d ever, with the ownship.

cdnas(ownid:Identifier,nas:(exists?(ownid))) : bool =
...

Formal Development of CDR2D

Flat Earth?

IMPORTING vectors@ECEF,
vectors@vect3_basis,
vectors@trackAngles_2D

% Radius of the earth
R : posreal = 6353000 % [m]

geo2sxy(lat_ref,lon_ref:real)(lat,lon:real) : Position =
LET midlat = (lat_ref+lat)/2,

midlon = (lon_ref+lon)/2,
refxyz = spherical2xyz(R,midlat,midlon),
pxyz = spherical2xyz(R,lat,lon) IN

sphere_to_2D_plane(refxyz)(pxyz)

gstrk2vxy(gs:posreal,trk:real): AbsVelocity =
v_from(trk,gs)

Formal Development of CDR2D

Proving cd2d ever Incompleteness

cd2d_ever_incomplete : THEOREM
NOT complete?(cd2d_ever)

Formal Development of CDR2D

Proving cd2d ever Incompleteness

cd2d_ever_incomplete :

{-1} complete?(cd2d_ever)
|-------

Rule? (expand "complete?")
Expanding the definition of complete?, this simplifies to:
cd2d_ever_incomplete :

{-1} FORALL (so, vo, si, vi):
cd2d_ever(so,vo,si,vi) IMPLIES conflict?(so-si,vo-vi)

|-------

Rule? (inst -1 "(-D-T,0)" "(2,0)" "(D,0)" "(1,0)")

Formal Development of CDR2D

Proving cd2d ever Incompleteness

Formal Development of CDR2D

cd2d_ever_incomplete.1 :

{-1} cd2d_ever((# x:=-D - T, y:=0 #), (# x:=2, y:=0 #),
(# x:=D, y:=0 #), (# x:=1, y:=0 #))

IMPLIES
conflict?((# x:=-D - T, y:=0 #) - (# x:=D, y:=0 #),

(# x:=2, y:=0 #) - (# x:=1, y:=0 #))
|-------

Rule? (grind :exclude "sq")
{-1} t!1 >= 0
{-2} t!1 <= T
{-3} sq(-D - D - T + t!1) < sq(D)
|-------

Formal Development of CDR2D

Formal Development of CDR2D

Rule? (rewrite "sq_neg" :dir rl)
Rewriting using sq_neg, matching in *, this simplifies to:
cd2d_ever_incomplete.1.1 :

[-1] t!1 >= 0
[-2] t!1 <= T
{-3} sq(-(-D - D - T + t!1)) < sq(D)
|-------

Rule? (rewrite "sq_lt")
Rewriting using sq_lt, matching in *,
This completes the proof of cd2d_ever_incomplete.1.1.

All the other sub-goals are easily discharged by grind,
decompose-equality, and assert.

Formal Development of CDR2D

Proving cd2d ever Correctness

cd2d_ever_correct : THEOREM
correct?(cd2d_ever)

Formal Development of CDR2D

Proving cd2d ever Correctness

After a few strategies, we have two sub-goals:

cd2d ever correct.1:
{-1} (so-si)*(vo-vi) < 0
[-2] sq((so-si+t*(vo-vi))‘x)+sq((so-si+t*(vo-vi))‘y) < sq(D)
|-------

{1} sq((so-si)*(vo-vi))+sq(D)*sqv(vo-vi) -
sqv(so-si)*sqv(vo-vi) > 0

cd2d ever correct.2:
[-1] sq((so-si+t*(vo-vi))‘x)+sq((so-si+t*(vo-vi))‘y) < sq(D)
|-------

{1} (so-si)*(vo-vi) < 0
{2} sqv(so-si) < sq(D)

How to prove these kinds of formulas? Very carefully :-).

Real Number Proving Tools

I Basic algebraic manipulations via Manip.

I Simplifications via Field.

I Interval arithmetic via Interval.

I Solving polynomial inequalities via Bernstein.

Real Number Proving Tools

Basic Manipulations via Manip

Basic Manipulations via Manip

I Manip is a package for algebraic manipulations of real-valued
expressions.

I http:
//shemesh.larc.nasa.gov/people/bld/manip.html.

I The package consists of:
I Strategies.
I Extended notations for formulas and expressions.
I Emacs extensions.
I Support functions for strategy developers.

I Manip is pre-installed in PVS 5.0.

http://shemesh.larc.nasa.gov/people/bld/manip.html
http://shemesh.larc.nasa.gov/people/bld/manip.html

Basic Manipulations via Manip

Examples
Manip

{-1} x + y <= z
|-------

...
Rule? (mult-by -1 "100")

Multiplying both sides of selected formulas by given term,
this simplifies to:
mp :

{-1} (x + y) * 100 <= z * 100
|-------

...

Real Number Proving Tools

Basic Manipulations via Manip

A Few Manip Strategies

Strategy Description

(swap-rel fnums) Swap sides and reverse relations
(swap! expr-loc) x ◦ y ⇒ y ◦ x
(group! expr-loc LR) (x ◦ y) ◦ z ⇒ x ◦ (y ◦ z)
(flip-ineq fnums) Negate and move inequalities
(split-ineq fnum) Split ≤ (≥) into < (>) and =

Real Number Proving Tools

Basic Manipulations via Manip

More Strategies

Strategy Description

(mult-by fnums term) Multiply formula by term
(div-by fnums term) Divide formula by term
(move-terms fnum L|R tnums) Move additive terms left and right
(isolate fnum L|R tnum) Isolate additive terms
(cross-mult fnums) Perform cross-multiplications
(factor fnums) Factorize formulas
(factor! expr-loc) Factorize terms
(mult-eq fnum fnum) Multiply equalities
(mult-ineq fnum fnum) Multiply inequalities

Real Number Proving Tools

Basic Manipulations via Manip

Extended Formula Notation

I Standard
I *: All formulas.
I -: All formulas in the antecedent.
I +: All formulas in the consequent.

I Extended (Manip strategies only)
I (^ n1...nk): All formulas but n1,...,nk
I (-^ n1...nk): All antecedent formulas but n1,...,nk
I (+^ n1...nk): All consequent formulas but n1,...,nk



Real Applications

Real Number Proving Tools

Basic Manipulations via Manip

Extended Expression Notation

I Term indexes:
I L,R: Left- or right-hand side of a formula.
I n: n-th term from left to right in a formula.
I -n: n-th term from right to left in a formula.
I *: All terms in a formula.
I (^ n1...nk): All terms in a formula but n1,...,nk.

I Location references:
I (! fnum LR i1...in): Term in formula fnum, Left- or

Right-hand side, at recursive path location i1...ik.

Real Number Proving Tools

Simplifications via Field

Simplifications via Field

I Field is a package for simplifications in the closed field of real
numbers.

I http://shemesh.larc.nasa.gov/people/cam/Field.
I The package consists of:

I The strategies grind-reals and field.
I Several extra-tegies.
I Field is pre-installed in PVS 5.0.

http://shemesh.larc.nasa.gov/people/cam/Field

Simplifications via Field

grind-reals

{-1} x * y >= 0
{-2} x > 0
|-------

{1} y >= 0
Rule? (grind-reals)
Rewriting with pos_times_ge

Applying GRIND-REALS,
Q.E.D.

field

{-1} vox > 0
{-2} s * s - D*D > D
{-3} s * vix * voy - s * viy * vox /= 0
{-4} ((s * s - D*D) * voy - D * vox * sqrt(s*s - D*D))/

(s * (vix * voy - vox * viy)) * s * vox /= 0
{-5} voy * sqrt(s * s - D*D) - D * vox /= 0
|-------

{1} (viy * sqrt(s * s - D*D) - vix * D) /
(voy * sqrt(s * s - D*D) - vox * D) =
(D*D - s * s) / (((s * s - D*D) * voy - D * vox *
sqrt(s * s - D*D)) /
(s * (vix * voy - vox * viy)) * s * vox) +
vix / vox

Rule? (field)
Q.E.D.

Simplifications via Field

Some Extra-tegies

Strategy Description

(cancel-by fnum term) Cancel a common term in a formula
(skoletin fnum) Skolemize let-in expressions
(skeep fnum) Skolemize with same variable names
(neg-formula fnum) Negate a formula
(add-formula fnum fnum) Add two formulas

Interval Arithmetic via Interval

Interval Arithmetic via Interval
http://shemesh.larc.nasa.gov/people/cam/Interval

I Interval is a package for interval analysis.
I The package consists of:

I The library interval arith, which presents a formalization
of interval analysis for real-valued functions including:
trigonometric functions, logarithm and exponential functions,
square root, absolute value, etc.

I The strategy numerical, which implements a provably correct
branch-and-bound interval analysis algorithm.

I Interval is part of the NASA PVS Libraries.

http://shemesh.larc.nasa.gov/people/cam/Interval

Interval Arithmetic via Interval

A Simple Problem

Prove that the turn rate of an aircraft with a bank angle of 35o is
greater than 3o per second.

IMPORTING interval_arith@strategies

g:posreal=9.8 %[m/s^2]

v:posreal=250*0.514 %[m/s]

tr(phi:(Tan?)): MACRO real = g*tan(phi)/v

tr_35 : LEMMA
3*pi/180 <= tr(35*pi/180)

Interval Arithmetic via Interval

numerical

tr_35 :

|-------
{1} 3 * pi / 180 <= g * tan(35 * pi / 180) / v

Rule? (numerical)
Evaluating formula using numerical approximations,
Q.E.D.

Special prize: Prove this lemma in any theorem prover different
from PVS (Note: pi is the mathematical irrational number π and
tan is the trigonometric function tan).

Interval Arithmetic via Interval

A Simple Property of Logarithms

G(x:real|x < 1): MACRO real = 3*x/2 - ln(1-x)

A_and_S : LEMMA
let x = 0.5828 in
G(x) > 0

Interval Arithmetic via Interval

A Simple Property of Logarithms

A_and_S :

|-------
{1} LET x = 0.5828 IN 3 * x / 2 - ln(1 - x) > 0

Rule? (numerical)
Evaluating formula using numerical approximations,
Q.E.D.

Special prize: Prove this lemma in any theorem prover different
from PVS (Note: ln is natural logarithm function).

Interval Arithmetic

{-1} x ## [| 0, 2 |]
|-------

{1} sqrt(x) + sqrt(3) < pi + 0.1

Rule? (numerical :vars "x")
Evaluating formula using numerical approximations,
Q.E.D.

Interval Analysis

Prove that for all x ∈ [−1
2 , 0],

| ln(1 + x)− x | − ε ≤ 2x2,

where ε = 0.15:1

ex_ba : LEMMA
x ## [|-1/2,0|] IMPLIES
abs(ln(1+x) - x) - epsilon <= 2*sq(x)

1Thanks to Behzad Akbarpour.

Interval Arithmetic via Interval

instint

ex_ba :
|-------

{1} FORALL (x: real):
x ## [|-1/2,0|] IMPLIES abs(ln(1+x)-x)-0.15 <= 2*sq(x)

Rule? (skeep)
ex_ba :
{-1} x ## [| -1 / 2, 0 |]
|-------

{1} abs(ln(1 + x) - x) - 0.15 <= 2 * sq(x)

Rule? (numerical :vars (("x" 10)))
Evaluating formula using numerical approximations,
Q.E.D.

Solving Polynomial Inequalities via Bernstein
http://shemesh.larc.nasa.gov/people/cam/Bernstein

I Bernstein is a package for solving multivariate polynomial
global optimization problems using Bernstein polynomials.

I The package consists of:
I The library Bernstein, which presents a formalization of an

efficient representation of multivariate polynomials.
I The strategy bernstein, which discharges simply quantified

multivariate polynomial inequalities on closed/open ranges.
I Grizzly, which is a prototype client-server tool for solving

global optimization problems.

I Bernstein is part of the NASA PVS Libraries.

http://shemesh.larc.nasa.gov/people/cam/Bernstein

Solving Polynomial Inequalities

IMPORTING Bernstein@strategy

p1 : LEMMA
FORALL (x,y:real): -0.5 <= x AND x <= 1 AND

-2 <= y AND y <= 1 IMPLIES
4*x^2-(21/10)*x^4+(1/3)*x^6+(x-3)*y-4*y^2+4*y^4 > -3.4

p2 : LEMMA
EXISTS (x,y:real): -0.5 <= x AND x <= 1 AND

-2 <= y AND y <= 1 AND
4*x^2-(21/10)*x^4+(1/3)*x^6+(x-3)*y-4*y^2+4*y^4 < -3.39

|-------
{1} FORALL (x, y: real):

-0.5 <= x AND x <= 1 AND -2 <= y AND y <= 1 IMPLIES
4*x^2-(21/10)*x^4+(1/3)*x^6+(x-3)*y-4*y^2+4*y^4 > -3.4

Rule? (bernstein)
Proving polynomial inequality using Bernstein’basis,
Q.E.D.

Special prize: Prove this lemma in any theorem prover different
from PVS

Solving Polynomial Inequalities via Bernstein

A Final Example

cd2d_numeric_conflict: LEMMA
-10 <= s‘x AND s‘x <= -8 AND
-10 <= s‘y AND s‘y <= -8 AND
6 <= v‘x AND v‘x <= 9 AND
6 <= v‘y AND v‘y <= 9 AND
D>=4 AND D<=6
IMPLIES
cd2d_ever?(s,v)

The proof is achieved with (grind), followed by (bernstein).

Concluding Remarks

Formal Methods in NextGen:

I NextGen is a system of systems: aircraft, physical
environment, human operators.

I Formal methods for system engineering rather than for
software engineering.

I Different sources of uncertainty.

I Highly distributed safety critical systems.

Concluding Remarks

Practical Challenges

I Evolutionary vs. revolutionary concepts.

I Theoretical vs. practical solutions.

I Local vs. global solutions.



Real Applications

Concluding Remarks

Current Technical Challenges

Automation, automation, automation:

I Non-linear arithmetic.

I Floating point arithmetic.

I Probabilistic reasoning.

I Numerical integration.

